PRÁCTICA 6:
Teoremas del Cálculo Vectorial

Teorema de Green

Sean D una región del tipo (III) y C su frontera. Supongamos que $P : D \to \mathbb{R}$ y $Q : D \to \mathbb{R}$ son de clase C^1. Entonces

$$\int_{C+} Pdx + Qdy = \iint_D \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right) dxdy$$



Teorema de Stokes

Supongamos que S es una superficie orientada definida por una parametrización $\overline{T} : D \subset \mathbb{R}^2 \to S$. Sea $C = \partial S$ la curva cerrada simple que constituye la frontera orientada de S y sea $\overline{F}$ un campo vectorial C^1 en S. Entonces se cumple

$$\iint_S \mathrm{rot}\,\overline{F} \cdot \overline{dS} = \int_{\partial S} \overline{F} \cdot \overline{ds}$$



Teorema de Gauss o teorema de la Divergencia

Supongamos que Ω es una región en el espacio del tipo (IV). Denotemos por $\partial\Omega$ la superficie cerrada orientada, según la normal exterior, que es la frontera de Ω. Sea $\overline{F}$ un campo vectorial C^1 definido en Ω. Entonces se cumple

$$\iint_{\partial\Omega} \overline{F} \cdot \overline{dS} = \iiint_\Omega (\mathrm{div}\,\overline{F})\, dxdydz$$



Al igual que en la práctica anterior, para calcular el producto escalar de dos vectores con *wxMaxima*, utilizaremos el punto del teclado **.** y para el producto vectorial utilizaremos la orden **~** (Alt Gr 4 del teclado) y la orden **express,** cargando previamente el paquete **load(vect).**

```
(%i2)   t1:[1,2,3];
        t2:[a,b,c];
(t1)    [1,2,3]
(t2)    [a,b,c]
(%i3)   pe:t1.t2;
(pe)    3 c+2 b+a
(%i4)   load(vect);
(%o4)   C:\maxima-5.38.1\share\maxima\5.38.1_5_gdf93b7b_dirty\share
(%i5)   pv:express(t1~t2);
(pv)    [2 c-3 b,3 a-c,b-2 a]
```

Para calcular la divergencia y el rotacional, una vez cargado el paquete **load(vect)**, el programa *wxMaxima* dispone de las ordenes **div** y **curl** para calcular la divergencia y el rotacional respectivamente. Veamos un ejemplo.

```
(%i6)   F:[x*y*z,x^2*z*y,z+y];
(F)     [x y z,x^2 y z,z+y]
(%i7)   div(F);
(%o7)   div([x y z,x^2 y z,z+y])
(%i8)   curl(F);
(%o8)   curl([x y z,x^2 y z,z+y])
```

Como se ve, *wxMaxima*, se limita a devolvernos las mismas órdenes que le introducimos. Para que nos dé los valores de la divergencia y del rotacional necesitamos las órdenes **express** y **ev**.

```
(%i10)  d:express(div(F));
        diver:ev(d,diff);
```

(d) $\frac{d}{d z}\left(z+y\right)+\frac{d}{d y}\left(x^2 y z\right)+\frac{d}{d x}\left(x y z\right)$

(diver) $y z+x^2 z+1$

```
(%i12)  r:express(curl(F));
        rot:ev(r,diff);
```

(r) $\left[\frac{d}{d y}\left(z+y\right)-\frac{d}{d z}\left(x^2 y z\right),\frac{d}{d z}\left(x y z\right)-\frac{d}{d x}\left(z+y\right),\frac{d}{d x}\left(x^2 y z\right)-\frac{d}{d y}\left(x y z\right)\right]$

(rot) $[1-x^2 y,x y,2 x y z-x z]$

Ejemplo 1

Aplicando el teorema de Green, calcular la integral de línea.

$$\oint_C y^3\,dx + x^3\,dy$$

En sentido positivo, siendo C la elipse $2\,x^2 + y^2 = 1$.

Solución con wxMaxima:

En este caso $P(x,y) = y^3$ y $Q(x,y) = x^3$ y mediante el teorema de Green, sabemos que se cumple:

$$\oint_C y^3\,dx + x^3\,dy = \iint_D \left(\frac{\partial Q(x,y)}{\partial x} - \frac{\partial P(x,y)}{\partial y}\right) dxdy$$

Por lo que calcularemos dicha integral doble.

```
(%i14) wxdraw2d(proportional_axes=xy,
       implicit(2*x^2+y^2=1,
       x,-1/sqrt(2),1/sqrt(2),y,-1,1));
```

(%t14)



(%o14)

```
(%i16) p(x,y):=y^3;
       q(x,y):=x^3;
```

(%o15) $p(x,y):=y^3$

(%o16) $q(x,y):=x^3$

```
(%i17) iterada1:integrate(
       (diff(q(x,y),x)-diff(p(x,y),y)),y,
       -sqrt(1-2*x^2),sqrt(1-2*x^2));
```

(iterada1) $\sqrt{1-2\,x^2}\left(5\,x^2-1\right)-\left(1-5\,x^2\right)\sqrt{1-2\,x^2}$

```
(%i18) integrate(iterada1, x, -1/sqrt(2), 1/sqrt(2));
```

(%o18) $-\frac{3\,\pi}{2^{7/2}}$

Ejemplo 2

Aplicar el teorema de Stokes para calcular la integral de línea.

$$\int_C 3ydx - xzdy + yz^2 dz$$

Siendo C la intersección del paraboloide $x^2 + y^2 = 2z$ y el plano $z = 2$.

Solución con wxMaxima:

Mediante el teorema de Stokes, sabemos que siendo $\vec{F} = 3y\,\vec{i} - xz\,\vec{j} + yz^2\,\vec{k}$ se cumple:

$$\int_C 3ydx + xzdy + yz^2 dz = \iint_S rot(\vec{F}) \cdot d\,\vec{S}$$

Por lo que, en primer lugar calcularemos $rot(\vec{F})$ y a continuación dicha integral de superficie .

(%i19)
```
wxdraw3d(
implicit(x^2+y^2=2*z,x,-2,2,y,-2,2,z,0,2));
```

(%t19)



(%o19)

```
-->      /* Cálculo del rotacional */
```

(%i20)
```
F:[3*y,-x*z,y*z^2];
```

(F) $[3\,y, -x\,z, y\,z^2]$

(%i22)
```
r:express(curl(F));
rot:ev(r,diff);
```

(r) $\left[\frac{d}{d\,y}\left(y\,z^2\right) - \frac{d}{d\,z}\left(-x\,z\right), \frac{d}{d\,z}\left(3\,y\right) - \frac{d}{d\,x}\left(y\,z^2\right), \frac{d}{d\,x}\left(-x\,z\right) - \frac{d}{d\,y}\left(3\,y\right)\right]$

(rot) $[z^2 + x, 0, -z-3]$

```
(%i23) define(rotacional(x,y,z),rot);
```
(%o23) $\mathrm{rotacional}(x,y,z):=[z^2+x,0,-z-3]$

```
--> /* Ecuaciones paramétricas en coordenadas polares */
```

```
(%i26) x(u,v):=u*cos(v);
       y(u,v):=u*sin(v);
       z(u,v):=u^2/2;
```
(%o24) $\mathrm{x}(u,v):=u\cos(v)$

(%o25) $\mathrm{y}(u,v):=u\sin(v)$

(%o26) $\mathrm{z}(u,v):=\frac{u^2}{2}$

```
--> /* Cálculo de la normal */
```

```
(%i30) t(u,v):=[x(u,v),y(u,v),z(u,v)];
       tu:diff(t(u,v),u);
       tv:diff(t(u,v),v);
       normal:express(tu~tv);
```
(%o27) $\mathrm{t}(u,v):=[\mathrm{x}(u,v),\mathrm{y}(u,v),\mathrm{z}(u,v)]$

(tu) $[\cos(v),\sin(v),u]$

(tv) $[-u\sin(v),u\cos(v),0]$

(normal) $[-u^2\cos(v),-u^2\sin(v),u\sin(v)^2+u\cos(v)^2]$

```
--> /* Cálculo de la función integrando */
```

```
(%i31) int:rotacional(x(u,v),y(u,v),z(u,v)).normal;
```
(int) $\left(-\frac{u^2}{2}-3\right)\left(u\sin(v)^2+u\cos(v)^2\right)-u^2\cos(v)\left(u\cos(v)+\frac{u^4}{4}\right)$

```
(%i32) int:trigsimp(int);
```
(int) $\frac{4u^3\sin(v)^2-u^6\cos(v)-6u^3-12u}{4}$

```
--> /* Cálculo de la integral */
```

```
(%i33) iterada1:integrate(int,u,0,2);
```
(iterada1) $\frac{112\sin(v)^2-128\cos(v)-336}{28}$

```
(%i34) integrate(iterada1, v, 0, 2*%pi);
```
(%o34) -20π

Ejemplo 3

Sea $\vec{F} = 2x\,\vec{i} + y^2\,\vec{j} + z^2\,\vec{k}$ y S la superficie esférica $x^2 + y^2 + z^2 = 1$. Aplicar el teorema de Gauss para calcular la integral de superficie.

$$\iint_S \vec{F} \cdot \vec{dS}$$

Solución con wxMaxima:

Mediante el teorema de Gauss, sabemos que siendo $\vec{F} = 2x\,\vec{i} + y^2\,\vec{j} + z^2\,\vec{k}$ se cumple:

$$\iint_S \vec{F} \cdot \vec{dS} = \iiint \operatorname{div}(\vec{F}) \cdot d\,\vec{S}$$

Por lo que, en primer lugar calcularemos $\operatorname{div}(\vec{F})$ y a continuación dicha integral triple .

```
(%i1)   kill(all)$

-->     /* Cálculo de la divergencia */

(%i1)   load(vect)$

(%i2)   F:[2*x,y^2,z^2];
```

(F) $[2\,x, y^2, z^2]$

```
(%i4)   d:express(div(F));
        diver:ev(d,diff);
```

(d) $\frac{d}{d\,z}\,z^2 + \frac{d}{d\,y}\,y^2 + \frac{d}{d\,x}\,(2\,x)$

(diver) $2\,z+2\,y+2$

```
(%i5)   define(divergencia(x,y,z),diver);
```

(%o5) $\mathrm{divergencia}(x,y,z):=2\,z+2\,y+2$

```
-->     /* Cambio de variable a esféricas */

(%i8)   x(r,t,fi):=r*sin(fi)*cos(t);
        y(r,t,fi):=r*sin(fi)*sin(t);
        z(r,t,fi):=r*cos(fi);
```

(%o6) $x(r,t,fi):=r\sin(fi)\cos(t)$

(%o7) $y(r,t,fi):=r\sin(fi)\sin(t)$

(%o8) $z(r,t,fi):=r\cos(fi)$

```
(%i11) m:jacobian([x(r,t,fi),y(r,t,fi),z(r,t,fi)],[r,t,fi]);
       j:determinant(m)$
       j:trigsimp(j);
```

$$(m)\quad \begin{bmatrix} \sin(fi)\cos(t) & -\sin(fi)\,r\sin(t) & \cos(fi)\,r\cos(t) \\ \sin(fi)\sin(t) & \sin(fi)\,r\cos(t) & \cos(fi)\,r\sin(t) \\ \cos(fi) & 0 & -\sin(fi)\,r \end{bmatrix}$$

$$(j)\quad -\sin(fi)\,r^2$$

```
-->    /* Cálculo de las integrales iteradas */
```

```
(%i12) iterada1:integrate(divergencia(x(r,t,fi),y(r,t,fi)
       ,z(r,t,fi))*j,r,0,1);
```

$$(iterada1)\quad -\frac{\sin(fi)\left(3\sin(fi)\sin(t)+3\cos(fi)+4\right)}{6}$$

```
(%i13) iterada2:integrate(iterada1,fi,0,%pi);
```

$$(iterada2)\quad -\frac{\frac{3\pi\sin(t)+5}{2}+\frac{11}{2}}{6}$$

```
(%i14) iterada3:integrate(iterada2,t,0,2*%pi);
```

$$(iterada3)\quad -\frac{8\pi}{3}$$

EJERCICIOS PROPUESTOS

1.

Aplicando el teorema de Green, calcular la integral de línea.

$$\oint_C -y^3dx+3xy^2dy$$

En sentido positivo, siendo C la elipse $2\,x^2+y^2+2x=0$.

Solución: $\frac{3\pi}{8\sqrt{2}}$

2.

Aplicar el teorema de Stokes para calcular la integral de línea.

$$\oint_C ydx-xdy+zdz$$

Siendo C la intersección del cono $z=\sqrt{x^2+y^2}$ y la esfera $x^2+y^2+z^2=2$.

Solución: -2π

3.

Sea $\overrightarrow{F} = x\,\overrightarrow{i} + y^2\,\overrightarrow{j} + z^2\,\overrightarrow{k}$ y S la superficie cerrada y limitada por los paraboloides $z = x^2 + y^2$ y $z = 1 - x^2 - y^2$.
Aplicar el teorema de Gauss para calcular la integral de superficie.

$$\iint_S \overrightarrow{F} \cdot \overrightarrow{dS}$$

Solución: $\boldsymbol{\frac{\pi}{2}}$

4.

Aplicar el teorema de Stokes para calcular la integral de línea.

$$\int_C xydx - xzdy + yzdz$$

Siendo C la intersección del paraboloide $z = 2 - x^2 - y^2$ y el plano $z = 1$.

Solución: $\boldsymbol{-\pi}$

5.

Sea $\overrightarrow{F} = 2x\,\overrightarrow{i} + y\,\overrightarrow{j} + z^2\,\overrightarrow{k}$ y S la superficie cerrada y limitada por la esfera $x^2 + y^2 + z^2 = 1$ y el cono $z = \sqrt{x^2 + y^2}$.
Aplicar el teorema de Gauss para calcular la integral de superficie.

$$\iint_S \overrightarrow{F} \cdot \overrightarrow{dS}$$

Solución: $\boldsymbol{2.625700532418671}$

6.

Aplicar el teorema de Stokes para hallar la integral de línea $\int_C (x^2+y^2+z^2)dx+(x^3+y^3+z^3)dy+(x^4+y^4+z^4)dz$, siendo C la intersección de las superficies:$z = 1 - \sqrt{x^2+y^2}$; $x^2 + y^2 + x = 0$.
a) Utilizando como parámetros las coordenadas cartesianas. b) Utilizando como parámetros las coordenadas polares. c) Comprobar que se obtiene el mismo resultado.
Nota: Representar gráficamente las superficies.

Solución: $\boldsymbol{1.180116287992763}$

7.

Aplicar el teorema de Stokes para hallar la integral de línea $\int_C zy^2dx + x^2zdy + xy^2dz$, siendo C la intersección de las superficies:$z = 2 - \sqrt{x^2+y^2}$; $x^2 + y^2 + 2y = 0$.
a) Utilizando como parámetros las coordenadas cartesianas. b) Utilizando como parámetros las coordenadas polares. c) Comprobar que se obtiene el mismo resultado.
Nota: Representar gráficamente las superficies.

Solución: $\boldsymbol{-2.81398966197822}$

8.

Aplicar el teorema de Gauss para hallar la integral de superficie $\iint_S \overrightarrow{F}\overrightarrow{dS}$, siendo $\overrightarrow{F} = (x^3, y^3, z^3)$ y siendo S la superficies cerrada y limitada por: $z = 4 - (x^2 + y^2)$; $x^2 + y^2 - 2y = 0$; $z = 0$.
a) Utilizando coordenadas cartesianas. b) Utilizando coordenadas cilíndricas. c) Comprobar que se obtiene el mismo resultado.
Nota: Representar gráficamente las superficies.

Solución: $\boldsymbol{\frac{125\pi}{4}}$